The outer automorphism f of the S_6 of order two with f(2, 3, 4, 5, 6) = (2, 3, 4, 5, 6)

```
> restart:with(group): el:=x->elements(x): r:=[[1,2]]:
  m:=[[2,3,4,5,6]]: s:=[[3,4,6,5]]:
  print('r'=r, 'm'=m, 's'=s, 's^2'=mulperms(s,s));
  S6:=permgroup(6,{r,m}):
  print('S6'=S6, 'order(S6)'=grouporder(S6));
```

$$r = [[1, 2]], m = [[2, 3, 4, 5, 6]], s = [[3, 4, 6, 5]], s^2 = [[3, 6], [4, 5]]$$

$$S6 = \text{permgroup}(6, \{[[2, 3, 4, 5, 6]], [[1, 2]]\}), \text{order}(S6) = 720$$

The elements r and m are the two classical generators of S_6. Repeated conjugation by m leads to the traditional generating set T of five transpositions, given by T = {(1,2), (1,3), (1,4), (1,5), (1,6)}.

```
> h:=x->mulperms(mulperms(invperm(m),x),m):T:={r,h(r),(h@h)(r),
  (h@@3)(r),(h@@4)(r)};
```

$$T := \{[[1, 6]], [[1, 5]], [[1, 2]], [[1, 3]], [[1, 4]]\}$$

```
> M:=permgroup(6,{m}); cM:=centralizer(S6,m);
  print('M_as_set'=el(M));
  print(evalb(M=cM),evalb(el(M)=el(M)));
```

$$M := \text{permgroup}(6, \{[[2, 3, 4, 5, 6]]\})$$

$$cM := \text{permgroup}(6, \{[[2, 6, 5, 4, 3]]\})$$

$$M\_as\_set = \{[\ ], [[2, 3, 4, 5, 6]], [[2, 6, 5, 4, 3]], [[2, 5, 3, 6, 4]], [[2, 4, 6, 3, 5]]\}$$

$$\mathit{false}, \mathit{true}$$

Apparently two permutation groups (of the same degree) are only equal 'as permgroups' if there sets of generators coincide.

```
>  nM:=normalizer(S6,M); el(Sylow(nM,5)); H:=el(nM):
  print('H'=H); grouporder(nM);
```

$$nM := \text{permgroup}(6, \{[[2, 3, 4, 5, 6]], [[3, 4, 6, 5]]\})$$

$$\{[\ ], [[2, 3, 4, 5, 6]], [[2, 6, 5, 4, 3]], [[2, 5, 3, 6, 4]], [[2, 4, 6, 3, 5]]\}$$

$$H = \{[[2, 6, 3, 4]], [\ ], [[2, 4], [5, 6]], [[2, 3, 5, 4]], [[2, 6, 4, 5]], [[2, 3, 4, 5, 6]], [[3, 4, 6, 5]], [[2, 5, 4, 6]], [[2, 3, 6, 5]], [[2, 5, 6, 3]], [[2, 6], [3, 5]], [[3, 6], [4, 5]], [[2, 3], [4, 6]], [[2, 6, 5, 4, 3]], [[2, 5], [3, 4]], [[2, 4, 5, 3]], [[2, 4, 3, 6]], [[2, 5, 3, 6, 4]], [[2, 4, 6, 3, 5]], [[3, 5, 6, 4]]\}$$

$$20$$

So the normalizer of M (of order 5) is the group H of order 20 and, since M is normal in H, it is the only Sylow 5-subgroup of H. In other words, there are no other 5-cykels in H than the four already contained in M, as you can verify above. We will now show that f(s) = s. Conjugation by s maps m onto m^2. That is $s^{(-1)}\, m\, s = m^2$.

```
> mulperms(mulperms(invperm(s),m),s);
```

$$[[2, 4, 6, 3, 5]]$$

Hence, since f is an automorphism with f(m) = m, we also have $f(s)^{(-1)}\, m\, f(s) = m^2$ and

$f(s)\,s^{(-1)} = c$, say, must belong to the centralizer of *m*, that is to *M* and $f(c) = c$. On the other hand, since f has order two by assumption, $f(c) = s\,f(s)^{(-1)} = c^{(-1)}$.Conclusion *c* must be the identity and $f(s) = s$. Moreover, since *r* = (1,2) commutes with *s*, so does the image $f(r)$, which is a product of 3 disjoint transpositions, since f is an *outer* automorphism.

```
> el(centralizer(S6,s));
```

{[ ], [[1, 2], [3, 5, 6, 4]], [[3, 4, 6, 5]], [[1, 2]], [[1, 2], [3, 4, 6, 5]],
[[3, 6], [4, 5]], [[1, 2], [3, 6], [4, 5]], [[3, 5, 6, 4]]}

Conclusion f((1,2)) = (1,2)(3,6)(4,5), or $f(r) = r\,s^2$. Then cyclic permutation, that is, repeated conjugation by *m*, gives us the other images under f of the transpositions in *A*. In fact, f((1,3)) = (1,3)(4,2)(5,6), f((1,4)) = (1,4)(5,3)(6,2), f((1,5) = (1,5)(6,4)(2,3) and f((1,6)) = (1,6)(2,5)(3,4). This is all we need to let Maple do the rest of the work. We start by a procedure to extend the known part of the graph of f, that is the set of all known ordered pairs [*x*, $f(x)$]. We use the fact that f is selfinverse.

```
> p:=proc(V::set,W::set)
      local T, v, w, X::set:
       X:=V union W;
        for v in V do for w in W do
  T:={[mulperms(v[1],w[1]),mulperms(v[2],w[2])],[mulperms(v[1],
  w[2]),mulperms(v[2],w[1])]}:
  X:=X union T:
        od: od: X;
    end:
```

A test seems appropriate. It is very important to write down the input in strict lexicographical order, as Maple does, to avoid multiple occurrences of the same permutation.

```
> A:={[[[1,2]],[[1,2],[3,6],[4,5]]],
      [[[1,3]],[[1,3],[2,4],[5,6]]],
      [[[1,4]],[[1,4],[2,6],[3,5]]],
      [[[1,5]],[[1,5],[2,3],[4,6]]],
      [[[1,6]],[[1,6],[2,5],[3,4]]]}:
  p({[[[1,2]],[[1,2],[3,6],[4,5]]]},{[[[1,3]],[[1,3],[2,4],[5,6
  ]]]});
```

{[[[1, 2]], [[1, 2], [3, 6], [4, 5]]], [[[1, 3]], [[1, 3], [2, 4], [5, 6]]],
[[[1, 4, 2, 3], [5, 6]], [[1, 2, 3, 6], [4, 5]]], [[[1, 2, 3]], [[1, 4, 6], [2, 3, 5]]]}

Let us now head for the complete graph. The procedure below will be used for a stopping criterion. Look at the example to see what it does.

```
> z:=proc(V::set)
      local v::list, X::set:
     X:={}:
      for v in V do X:=X union {v[1]} od:
     X;
  end:
  z(A);
```

{[[1, 6]], [[1, 5]], [[1, 2]], [[1, 3]], [[1, 4]]}

```
> q:=proc(A::set,n::posint)
      local S::set,T::set, W::set;
       S:={[[],[]]}: W:={[]}:
        while nops(W)<n do S:=S union p(S,A): W:=z(S): od:
       S;
      end:
```

You might wish to use a small value of *n* first, as a testcase, Do not use a value of *n* greater than the order of the group, 720 in this case, for the procedure might run forever. The next execution is the only one which takes noticeable time, between 10 and 20 seconds, say.

```
> graph:=q(A,720):nops(graph);
```

$$720$$

It appears that we have found all ordered pairs [x, f(x)] and we may immitate a function f accordingly.

```
> for v in graph do f(v[1]):=v[2] od:
  print('f([[1,2]])'=f([[1,2]]),'f([[3,4,6,5]])'=f([[3,4,6,5]])
  );
```

$$f([[1,2]]) = [[1,2],[3,6],[4,5]],\ f([[3,4,6,5]]) = [[3,4,6,5]]$$

Maple will now show that *H* is exactly the subgroup of all elements unaltered by f.

```
> fix:=proc(V::set)
    local v::list,X::set:
      X:={}:
     for v in V do if v[1]=v[2] then X:=X union {v[1]} fi od:
      X;
  end:
> fix(graph);evalb(H=fix(graph));
```

$$\{[[2,6,3,4]], [\ ], [[2,4],[5,6]], [[2,3,5,4]], [[2,6,4,5]], [[2,3,4,5,6]], [[3,4,6,5]], [[2,5,4,6]], [[2,3,6,5]], [[2,5,6,3]], [[2,6],[3,5]], [[3,6],[4,5]], [[2,3],[4,6]], [[2,6,5,4,3]], [[2,5],[3,4]], [[2,4,5,3]], [[2,4,3,6]], [[2,5,3,6,4]], [[2,4,6,3,5]], [[3,5,6,4]]\}$$

$$true$$

If c(x) is defined as f(x) $x^{(-1)}$ then f maps c(x) onto its inverse and hence the same holds for all powers of c(x), Note that $x^{(-1)}$ f(x) is the inverse of c($x^{(-1)}$). Maple will show that the converse also holds.

```
> c:=x->mulperms(f(x),invperm(x)):
  C:=proc(V::set)
      local v,t::list, X::set:
     X:={}:
    for v in V do t:=mulperms(v[2],invperm(v[1])): X:=X union
  {t} od:
     X;
  end:
> C(graph) intersect H; nops(%); C(graph) minus H; nops(%);
```

$$\{[\ ], [[2,4],[5,6]], [[2,6],[3,5]], [[3,6],[4,5]], [[2,3],[4,6]], [[2,5],[3,4]]\}$$

6

{[[1, 5, 3, 6, 4]], [[1, 6, 4, 2, 5]], [[1, 6], [2, 3, 5, 4]], [[1, 3, 5, 2, 4]],
[[1, 2, 3, 4, 5]], [[1, 5, 4, 3, 2]], [[1, 4], [2, 3, 6, 5]], [[1, 4], [2, 5, 6, 3]],
[[1, 4, 5, 6, 2]], [[1, 2, 6, 5, 4]], [[1, 2], [3, 5, 6, 4]], [[1, 4, 2, 5, 3]],
[[1, 5], [2, 4, 3, 6]], [[1, 6, 3, 5, 2]], [[1, 3, 2, 6, 5]], [[1, 2], [3, 4, 6, 5]],
[[1, 6], [2, 4, 5, 3]], [[1, 6, 2, 3, 4]], [[1, 5, 2, 4, 6]], [[1, 5], [2, 6, 3, 4]],
[[1, 3], [2, 5, 4, 6]], [[1, 6, 5, 4, 3]], [[1, 3, 4, 5, 6]], [[1, 4, 3, 2, 6]], [[1, 5, 6, 2, 3]],
[[1, 4, 6, 3, 5]], [[1, 2, 4, 6, 3]], [[1, 3], [2, 6, 4, 5]], [[1, 2, 5, 3, 6]], [[1, 3, 6, 4, 2]]
}

30

So we only have to show that the number of elements mapped onto their inverse by f is 36.

```
> Inv:=proc(V::set)
    local v,t::list, X::set:
      X:={}:
  for v in V do if mulperms(v[1],v[2])=[] then X:=X union
  {v[1]} fi od:
      X;
  end:
> Inv(graph):nops(%); evalb(Inv(graph)=C(graph));
```

36

true

Actually we only need the elements of order 3 to yield all 30 elements of C(*graph*) \ *H*.

```
> three1:=select(x -> evalb(grouporder(permgroup(6,{x[1]}))=3
  and nops(x[1])=1),graph):
  three2:=select(x -> evalb(grouporder(permgroup(6,{x[1]}))=3
  and nops(x[1])=2),graph):
  print(nops(three1),nops(three2));
```

40, 40

```
> C(three1) intersect C(three2); nops(C(three1)union
  C(three2));
```

{[[1, 5, 3, 6, 4]], [[1, 6, 4, 2, 5]], [[1, 3, 5, 2, 4]], [[1, 2, 3, 4, 5]], [[1, 5, 4, 3, 2]],
[[1, 4, 5, 6, 2]], [[1, 2, 6, 5, 4]], [[1, 4, 2, 5, 3]], [[1, 6, 3, 5, 2]], [[1, 3, 2, 6, 5]],
[[1, 6, 2, 3, 4]], [[1, 5, 2, 4, 6]], [[1, 6, 5, 4, 3]], [[1, 3, 4, 5, 6]], [[1, 4, 3, 2, 6]],
[[1, 5, 6, 2, 3]], [[1, 4, 6, 3, 5]], [[1, 2, 4, 6, 3]], [[1, 2, 5, 3, 6]], [[1, 3, 6, 4, 2]]}

30

three1 consists of 20 pairs [*x*, f(*x*)] where *x* is a 3-cycle without 1 and 20 pairs where the 3-cycle *x* does contain a 1, which behave completely differently if *c* is applied to *x*.

```
> three1no1:=select(x->not(member(1,convert((x[1])[1],set))),th
  ree1):z(%);
```

{[[3, 6, 5]], [[2, 4, 3]], [[2, 5, 3]], [[2, 4, 5]], [[2, 6, 3]], [[4, 5, 6]], [[2, 5, 6]],
[[2, 3, 6]], [[3, 4, 6]], [[2, 6, 4]], [[3, 6, 4]], [[3, 5, 4]], [[2, 4, 6]], [[4, 6, 5]],

[[2, 6, 5]], [[3, 5, 6]], [[2, 5, 4]], [[2, 3, 4]], [[3, 4, 5]], [[2, 3, 5]]}

```
> C(three1no1);
```

{[[1, 4], [2, 3, 6, 5]], [[1, 2], [3, 5, 6, 4]], [[1, 6], [2, 4, 5, 3]], [[1, 5], [2, 6, 3, 4]],
[[1, 3], [2, 5, 4, 6]]}

Five elements, *c* is a perfect four to one mapping on the set of 3-cycles without 1. These four elements mapped under c onto the same image are related through conjugation by the four powers of the 4-cycle in *H* contained in the image.

```
> c([[6,4,3]]);
Error, (in simpl/max) arguments must be of type algebraic
```

A deliberate error, as a reminder that the input should be in lexicographic order.

```
> c([[3,6,4]]);
  four:=x->mulperms(mulperms(invperm([[2,3,6,5]]),x),[[2,3,6,5]
  ]):
  'f([[2,3,6,5]])'=f([[2,3,6,5]]);
  print([[3,6,4]],four([[3,6,4]]),(four@four)([[3,6,4]]),(four@
  @3)([[3,6,4]]));
```

[[1, 4], [2, 3, 6, 5]]

f([[2, 3, 6, 5]]) = [[2, 3, 6, 5]]

[[3, 6, 4]], [[4, 6, 5]], [[2, 4, 5]], [[2, 3, 4]]

```
> print(c([[3,6,4]]),c([[4,6,5]]),c([[2,4,5]]),c([[2,3,4]]));
```

[[1, 4], [2, 3, 6, 5]], [[1, 4], [2, 3, 6, 5]], [[1, 4], [2, 3, 6, 5]], [[1, 4], [2, 3, 6, 5]]

For the inverses we proceed as follows

```
> print(f([[3,6,4]]),f([[4,6,5]]),f([[2,4,5]]),f([[2,3,4]]));
  print(c([[1,3,4],[2,6,5]]),c([[1,6,4],[2,3,5]]),c([[1,5,4],[2
  ,3,6]]),c([[1,2,4]],[3,6,5]));
```

[[1, 3, 4], [2, 6, 5]], [[1, 6, 4], [2, 3, 5]], [[1, 5, 4], [2, 3, 6]], [[1, 2, 4], [3, 6, 5]]

[[1, 4], [2, 5, 6, 3]], [[1, 4], [2, 5, 6, 3]], [[1, 4], [2, 5, 6, 3]], [[1, 6, 5, 4, 3]]

Finally the other 3-cycles, which contain 1, are all mapped by *c* onto the 5-cycles we have met before.

```
> C(three1 minus three1no1);evalb(C(three1 minus
  three1no1)=(C(three1) intersect C(three2)));
```

{[[1, 5, 3, 6, 4]], [[1, 6, 4, 2, 5]], [[1, 3, 5, 2, 4]], [[1, 2, 3, 4, 5]], [[1, 5, 4, 3, 2]],
[[1, 4, 5, 6, 2]], [[1, 2, 6, 5, 4]], [[1, 4, 2, 5, 3]], [[1, 6, 3, 5, 2]], [[1, 3, 2, 6, 5]],
[[1, 6, 2, 3, 4]], [[1, 5, 2, 4, 6]], [[1, 6, 5, 4, 3]], [[1, 3, 4, 5, 6]], [[1, 4, 3, 2, 6]],
[[1, 5, 6, 2, 3]], [[1, 4, 6, 3, 5]], [[1, 2, 4, 6, 3]], [[1, 2, 5, 3, 6]], [[1, 3, 6, 4, 2]]}

true

For compact display we may restrict the graph of f to a domain consisting of a complete set of right coset representatives of *H*. The set of representatives is produced by Maple without 5-cycles, 6-cycles or products of three disjoint transpositions and will subsequently be split up in sets of mutually conjugate elements.

```
> cosH:=cosets(S6,nM):nops(%);select(x->member(x[1],cosH),graph
  );
```

{[[[5, 6]], [[1, 3], [2, 6], [4, 5]]], [[[1, 2, 3], [4, 6]], [[1, 2, 6, 5, 4, 3]]],
[[[1, 2, 3], [4, 5]], [[1, 3, 6, 2, 4, 5]]], [[[1, 2, 3], [5, 6]], [[1, 5, 6, 3, 4, 2]]],
[[[1, 5, 6]], [[1, 2, 4], [3, 5, 6]]], [[[1, 2, 3], [4, 5, 6]], [[1, 6, 4]]],
[[[1, 5, 6, 4]], [[1, 6, 5, 2]]], [[[4, 5]], [[1, 2], [3, 4], [5, 6]]],
[[[4, 6, 5]], [[1, 6, 4], [2, 3, 5]]], [[[1, 4, 5, 6]], [[2, 3, 6, 4]]],
[[[1, 6], [4, 5]], [[1, 5], [2, 6]]], [[[1, 5, 4]], [[1, 3, 6], [2, 5, 4]]],
[[[4, 6]], [[1, 5], [2, 4], [3, 6]]], [[[4, 5, 6]], [[1, 4, 6], [2, 5, 3]]],
[[[1, 6, 5]], [[1, 4, 2], [3, 6, 5]]], [[[1, 6, 5, 4]], [[2, 4, 6, 3]]],
[[[1, 3], [4, 6, 5]], [[1, 5, 4, 3, 6, 2]]], [[[1, 3], [4, 5]], [[1, 4], [2, 3]]],
[[[1, 2, 3], [4, 6, 5]], [[2, 5, 3]]], [[[1, 5], [4, 6]], [[2, 6], [3, 4]]],
[[[1, 4, 6, 5]], [[1, 2, 5, 6]]], [[[1, 3], [4, 6]], [[1, 6], [3, 5]]],
[[[1, 6, 4, 5]], [[1, 3, 5, 4]]], [[[1, 4, 6]], [[1, 3, 2], [4, 6, 5]]],
[[[1, 4], [5, 6]], [[1, 5], [3, 4]]], [[[1, 3]], [[1, 3], [2, 4], [5, 6]]],
[[[1, 4]], [[1, 4], [2, 6], [3, 5]]], [[[1, 5]], [[1, 5], [2, 3], [4, 6]]],
[[[1, 3], [4, 5, 6]], [[1, 2, 6, 3, 4, 5]]], [[[1, 6]], [[1, 6], [2, 5], [3, 4]]],
[[[1, 3], [5, 6]], [[2, 5], [4, 6]]], [[[1, 4, 5]], [[1, 6, 3], [2, 4, 5]]],
[[[1, 2, 3]], [[1, 4, 6], [2, 3, 5]]], [[[1, 6, 4]], [[1, 2, 3], [4, 5, 6]]], [[ ], [ ]],
[[[1, 5, 4, 6]], [[1, 4, 5, 3]]]}

As promised, let us split *cosH* [forget the identity] into sets of elements of the same cycle structure. Notice the near absence of the number 2, the common element of *r* and *m*.

```
> select(x->grouporder(permgroup(6,{x}))=4,cosH);
```

{[[1, 4, 6, 5]], [[1, 5, 4, 6]], [[1, 4, 5, 6]], [[1, 5, 6, 4]], [[1, 6, 5, 4]], [[1, 6, 4, 5]]}

```
> select(x->grouporder(permgroup(6,{x}))=6,cosH);
```

{[[1, 3], [4, 5, 6]], [[1, 2, 3], [4, 5]], [[1, 2, 3], [5, 6]], [[1, 2, 3], [4, 6]],
[[1, 3], [4, 6, 5]]}

```
> select(x->grouporder(permgroup(6,{x}))=2 and nops(x)=1,cosH);
```

{[[1, 6]], [[5, 6]], [[1, 5]], [[4, 6]], [[1, 3]], [[4, 5]], [[1, 4]]}

```
> select(x->grouporder(permgroup(6,{x}))=2 and nops(x)=2,cosH);
```

{[[1, 5], [4, 6]], [[1, 6], [4, 5]], [[1, 3], [4, 6]], [[1, 3], [4, 5]], [[1, 3], [5, 6]],
[[1, 4], [5, 6]]}

```
> select(x->grouporder(permgroup(6,{x}))=3 and nops(x)=1,cosH);
```

{[[4, 5, 6]], [[1, 5, 4]], [[4, 6, 5]], [[1, 5, 6]], [[1, 4, 5]], [[1, 6, 5]], [[1, 6, 4]],
[[1, 4, 6]], [[1, 2, 3]]}

```
> select(x->grouporder(permgroup(6,{x}))=3 and nops(x)=2,cosH);
```

{[[1, 2, 3], [4, 5, 6]], [[1, 2, 3], [4, 6, 5]]}